#

pkgname=exo
pkgver=0.6.1
pkgrel=1
pkgdesc="Extensions to Xfce by os-cillation"
arch=('ppc')
license=('GPL2' 'LGPL')
url="http://www.xfce.org/projects/exo"
groups=('xfce4')
depends=("libxfce4util>=4.8.1" 'gtk2' 'hicolor-icon-theme' 'libnotify')
makedepends=('perl-uri' 'intltool' 'xorg-iceauth' 'pkg-config')
optdepends=('perl-uri: for mail-compose helper script')
options=('!libtool')
install=${pkgname}.install
source=(http://archive.xfce.org/src/xfce/${pkgname}/0.6/${pkgname}-${pkgver}.tar.bz2)
md5sums=('2ae3d9a9dedc3adebb748fa8724e02ce')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  # python2 fix
  export PYTHON=python2
  ./configure --prefix=/usr \
	--sysconfdir=/etc \
	--libexecdir=/usr/lib/xfce4 \
	--localstatedir=/var  \
	--disable-static \
	--enable-gio-unix \
	--enable-nls \
	--enable-notifications \
	--disable-python \
	--disable-hal \
	--disable-gtk-doc \
	--disable-gtk-doc-html \
	--without-html-dir
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
}
